# Is this integration error from a misuse of Mathematica or is it my poor math skills?

I have input this:

Integrate[1, f[t]*Exp[(v/V)*t]]


That is $\large\int 1*d(f(t)*e^{\frac{v}{V}*t})$

I expected this output: $\large f(t)*e^{\frac{v}{V}*t} + C$

but instead I get:

Integrate::ilim: "Invalid integration variable or limit(s) in E^((t\v)/V)\ f[t]. "


What am I doing wrong ?

PS: in the model at the origin of my troubles t is the only variable v and V are constants

-
But I want to integrate 1 with respect to f(t)*e^(v/V*t). –  Ouistiti Jul 31 '13 at 12:20
You can want whatever you like, but you have to adhere to the Mathematica syntax. It expects a variable in that position and what you provide is clearly not a variable. –  Sjoerd C. de Vries Jul 31 '13 at 12:23
Integrate[1, u] /. u -> f[t]*Exp[(v/V)*t] –  gpap Jul 31 '13 at 12:28
This works thanks gpap do you want to put it as an answer? –  Ouistiti Jul 31 '13 at 12:37
The type of integral you want to calculate often pops up when using the integration by parts trick. The resulting expression with a function behind the d is considered by many a (slight) abuse of notation and not mathematically solid. –  Sjoerd C. de Vries Jul 31 '13 at 12:39

Well, here's a (trivial) answer that works:

Module[{u}, Integrate[1, u] /. u -> f[t]*Exp[(v/V)*t]]

-
FYI: For best style u should be \[FormalU] or localized with Module. (+1) –  Mr.Wizard Jul 31 '13 at 15:39
agreed and amended –  gpap Jul 31 '13 at 15:47

Here's a way to implement your concept. But you do have to specify which variable is the variable of integration:

SetAttributes[int, HoldAll];
int[integrand_, D[measure_, var_Symbol]] := Integrate[integrand D[measure, var], var]

int[1, D[f[t]*Exp[(v/V)*t], t]]
(* E^((t v)/V) f[t] *)


And

int[f[t] Exp[(v/V) t], D[f[t]*Exp[(v/V)*t], t]]
(* 1/2 E^((2 t v)/V) f[t]^2 *)


Etc.

-